X olimpiada

Trečiojo etapo pirmoji dalis

Jaunesniųjų grupė

156. Ar daug kalbate telefonu? Norint išspręsti šį uždavinį tereikia mokėti dirbti su simbolių eilutėmis ir atlikti paprasčiausią paiešką. Uždavinys turi ir porą „kabliukų“: galima skambinti ir nemokamais numeriais, be to vieną parą prasidėjęs pokalbis gali baigtis kitą parą.

Ar pavyks parašyti trumpą programą priklausys nuo to, ar mokėsime pasirinkti tinkamas duomenų struktūras. Duomenims apie pokalbius saugoti patogu sukurti įrašų masyvą. Įrašo laukuose turėtų būti saugomi visi duomenys apie konkretų skambutį. Kadangi reikės ieškoti numerių su kuriais prakalbėta daugiausiai pinigų, susidarome ir kitą įrašų masyvą, kuriame bus saugoma informacija apie kiekvieną numerį bei prakalbėtą sumą.

const MAX_N = 100; { maksimalus pokalbių skaičius }

type skambutis = record

                     num: string [15]; { numeris }

                     dien,             { diena }

                     pr,            { pradžia nurodoma minutėmis}

                     trukmė: integer; { pokalbio trukmė minutėmis }

                     kaina: longint { kaina nurodoma centais}

                   end;

       pokalbiai = array [1..MAX_N] of skambutis;

       numeriai = array [1..MAX_N] of record

                                        num: string[15];

                                        kaina: longint

                                      end;



Skirtingai nuo kitų uždaviniu, pradinių duomenų įvedimas yra svarbi šio uždavinio dalis. Todėl pateikiame ir pradinių duomenų įvedimo procedūrą:

const PRAD = 'TELE.DAT';

procedure skaityti (var n: integer; var sk: pokalbiai);

{ perskaito pradinius duomenis }

  var f: text;

      i, valan, min, klaida: integer;

      litai, centai: longint;

      st1: string[1];

      st2: string [2];

      st3: string [3];

begin

  assign (f, PRAD);

  reset (f);

  readln (f, n);

  for i := 1 to n do

   begin

     read (f, sk[i].dien);

     read (f, st3); { _:_ }

     read (f, st2); { pokalbio pradžios valanda }

     val (st2, valan, klaida);

     read (f, st1); { taškas, skiriantis, valandas ir minutes }

     read (f, min);

     sk[i].pr := valan * 60 + min; { radome pokalbio pradžios laiką }

     read (f, st3); { _:_ }

     read (f, sk[i].num);  { perskaitomas numeris }

     while sk [i].num [1] = '*' do

       delete(sk[i].num, 1, 1);

     read (f, st3); { _:_ }

     read (f, sk [i].trukmė);

     read (f, st3); { _:_ }

     read (f, st3); { sumokėti litai }

     val(st3, litai, klaida);

     read (f, st1); { taškas, skiriantis litus ir centus }

     read (f, centai);

     sk[i].kaina := litai * 100 + centai;

   end;

  close (f);

end; { skaityti }



Perskaitę pradinius duomenis ieškosime reikiamos informacijos. Pirmiausia rasime penkis telefono numerius su kuriais prakalbėtos didžiausios pinigų sumos. Tam pirmiausia suskaičiuojame kiek buvo prakalbėta kiekvienu numeriu. Kadangi galėjo būti kelis kartus skambinta tuo pačiu numeriu, tai peržiūrint duomenis apie kiekvieną skambutį būtina patikrinti ar anksčiau nebuvo skambinta tuo numeriu ir jei taip – padidinti tuo numeriu prakalbėtą pinigų sumą.

Norėdami rasti kuriuo numeriu prakalbėta daugiausia pinigų reikia peržiūrėti duomenis apie visus numerius ir išrinkti numerį su didžiausia suma. Atspausdinę tą numerį jo prakalbėtą sumą prilyginame nuliui ir tokiu pat būdu ieškome antro numerio kuriuo prakalbėta daugiausia pinigų.

const NUM = 5; { reikia rasti penkis numerius }

procedure rasti_num (n: integer; sk: pokalbiai);

{ suranda penkis numerius, kuriais prakalbėta daugiausiai }

{ ir juos išspausdina }

  var kiek, max_sk, nr, i, ii, iii: integer;

      buvo: boolean;

      nums: numeriai; { duomenys apie pinigus, prakalbėtus kiekvienu }

                      { numeriu }

begin

  { apskaičiuojame kiek pinigų prakalbėta kiekvienu numeriu }

  max_sk := 0; { maksimalus numerių skaičius }

  for i := 1 to n do

    begin

      buvo := false; { ar buvo toks numeris }

      for ii := 1 to max_sk do

        if nums[ii].num = sk[i].num

           then begin

                  buvo := true;

                  nr := ii

                end;

      if buvo { jei toks numeris jau buvo }

        then nums[nr].kaina := nums[nr].kaina + sk[i].kaina

        else begin { įtraukiame numerį į sąrašą }

               max_sk := max_sk + 1;

               nums[max_sk].num := sk[i].num;

               nums[max_sk].kaina := sk[i].kaina

             end;

    end;

  { randame penkis numerius, kuriais prakalbėta daugiausia }

  kiek := NUM; { tiek numerių reikia rasti }

  if kiek > max_sk { jei nesusidaro tiek numerių }

     then kiek := max_sk;

  for i := 1 to kiek do

    begin

      ii := 1;

      for iii := 2 to max_sk do

        if nums[iii].kaina > nums [ii].kaina

           then ii := iii;

      if nums[ii].kaina <> 0

         then write(nums[ii].num, ' ');

      nums [ii].kaina := 0;

    end;

  writeln;

end; { rasti_num }



Norėdami rasti tris paras kada kalbėta ilgiausiai, pirma suskaičiuojame kiek prakalbėta kiekvieną mėnesio parą. Reikia nepamiršti, kad pokalbis gali prasidėti vieną parą, o baigtis – kitą. Turėdami duomenis apie kiekvienos paros pokalbių trukmę tokiu pat būdu kaip ir ieškodami penkių numerių randame tris paras kada kalbėta daugiausiai.

const PAROS = 3; { reikia rasti tris paras }

procedure rasti_paras (n: integer; sk: pokalbiai);

{ randa ir išspausdina tris paras, kada kalbėta daugiausia }

  var dp: array [1..31] of integer; { kiek kalbėta kiekvieną dieną }

      i, ii, iii: integer;

begin

  { randame kiek kalbėta kiekvieną parą }

  for i := 1 to 31 do

    dp[i] := 0;

  for i := 1 to n do

    begin

      if (sk[i].pr + sk[i].trukmė-1) < 24*60

         { jei pokalbis prasidėjo ir baigėsi tą pačią parą }

         then dp[sk[i].dien] := dp[sk[i].dien] + 

                                sk[i].trukmė

         else begin { jei pokalbis baigėsi kitą parą }

                dp[sk[i].dien] := 24*60 - sk[i].pr;

                dp[sk[i].dien+1] := sk[i].pr + 

                                    sk[i].trukmė - 24*60

              end;

    end;

  { randame tris ar mažiau paras, kuriomis kalbėta daugiausiai }

  for i := 1 to PAROS do

    begin

      ii := 1;

      for iii := 1 to 31 do

        if dp[iii] > dp[ii]

           then ii := iii;

      if dp[ii] > 0

         then write (ii, ' ');

      dp[ii] := 0

    end;

  writeln;

end; { rasti_paras }



Galiausiai randame išlaidų dalį kurią sudaro vietiniai ir minutės trukmės pokalbiai.

procedure viet_min (n: integer; sk: pokalbiai);

{ randa išlaidų dalį, kurią sudaro vietinių ir minutės }

{ trukmės pokalbiai }

  var i, viet, min, viso: longint;

begin

  { randama išlaidų dalis, kurią sudaro vietiniai pokalbiai }

  viet := 0; viso := 0;

  for i := 1 to n do

    begin

      if sk[i].num [1] <> '8'

         then viet := viet + sk[i].kaina;

      viso := viso + sk[i].kaina;

    end;

  write (viet/viso*100: 4: 2, ' ');

  { randama išlaidų dalis, kurią sudaro pokalbiai po viena minutę }

  min := 0;

  for i := 1 to n do

    if sk[i].trukmė = 1

       then min := min + sk[i].kaina;

  writeln (min/viso*100: 4: 2);

end; { viet_min }



Apjungę visas procedūras gausime uždavinį sprendžiančią programą:

program ar_daug_kalbate_telefonu;



const MAX_N = 100; { maksimalus pokalbių skaičius }

type skambutis = record

                     num: string [15]; { numeris }

                     dien,             { diena }

                     pr,            { pradžia nurodoma minutėmis}

                     trukmė: integer; { pokalbio trukmė minutėmis }

                     kaina: longint { kaina nurodoma centais}

                   end;

       pokalbiai = array [1..MAX_N] of skambutis;

       numeriai = array [1..MAX_N] of record

                                        num: string[15];

                                        kaina: longint

                                      end;



const PRAD = 'TELE.DAT';

procedure skaityti (var n: integer; var sk: pokalbiai);

{ perskaito pradinius duomenis }

  var ...

begin

  ...

end; { skaityti }



const NUM = 5; { reikia rasti penkis numerius }

procedure rasti_num (n: integer; sk: pokalbiai);

{ suranda penkis numerius, kuriais prakalbėta daugiausiai }

{ ir juos išspausdina }

  var ...

begin

  ...

end; { rasti_num }



const PAROS = 3; { reikia rasti tris paras }

procedure rasti_paras (n: integer; sk: pokalbiai);

{ randa ir išspausdina tris paras, kada kalbėta daugiausia }

  var ...

begin

  ...

end; { rasti_paras }



procedure viet_min (n: integer; sk: pokalbiai);

{ randa ir išspausdina išlaidų dalį, kurią sudaro vietinių ir minutės }

{ trukmės pokalbiai }

  var ...

begin

  ...

end; { viet_min }



var  n: integer;

     sk: pokalbiai;

begin { pagrindinė programos dalis }

  skaityti (n, sk);

  rasti_num (n, sk);

  rasti_paras (n, sk);

  viet_min (n, sk);

end.

�

Testo nr.�Pradiniai duomenys�Rezultatai�Paaiškinimai��1�10

10 : 21.50 : **8372585741252 : 030 : 153.00

11 : 14.10 : *********466484 : 125 : 008.75

14 : 15.11 : *********427801 : 001 : 000.07

16 : 03.10 : *********765859 : 015 : 001.05

18 : 07.00 : *********454555 : 021 : 001.47

18 : 11.11 : ******890044406 : 102 : 025.00

18 : 22.15 : ******827201224 : 001 : 001.56

21 : 13.30 : *********748945 : 046 : 003.22

25 : 13.35 : *********454555 : 240 : 016.80

29 : 14.01 : *********223322 : 122 : 008.54�8372585741252 890044406 454555 466484 223322

25 11 18

18.18 0.74�Atsitiktinai parinktas testas ��2�6

10 : 21.50 : *******82122221 : 030 : 015.00

11 : 14.10 : *******82122222 : 125 : 062.50

14 : 15.11 : *******82122223 : 011 : 005.50

16 : 03.10 : *******82122224 : 015 : 007.50

18 : 07.00 : *******82122225 : 021 : 010.05

19 : 11.11 : ******882122226 : 102 : 051.00�82122222 882122226 82122221 82122225 82122224

11 19 10

0.00 0.00�Nėra vienos minutės pokalbių, nėra vietinių pokalbių��3�6

10 : 21.50 : *********233222 : 030 : 002.10

11 : 14.10 : *********767676 : 125 : 008.75

14 : 15.11 : *************02 : 005 : 000.00

16 : 03.10 : *********767676 : 001 : 000.07

18 : 07.00 : *********333333 : 021 : 001.47

19 : 11.11 : *********222222 : 102 : 008.14�767676 222222 233222 333333

11 19 10

100.00 0.34�Buvo kalbėta nemokamu telefonu, todėl nesusidaro penki numeriai su kuriais būtų prakalbėta nors truputis pinigų.��4�2

10 : 21.50 : *********455555 : 030 : 002.09

11 : 14.10 : *********455555 : 001 : 000.07�455555

10 11

100.00 3.24�Iš viso buvo tik du skambučiai, kalbėta tik vienu numeriu, nėra tarpmiestinių pokalbių..��5�6

10 : 21.50 : ******890044444 : 030 : 153.00

11 : 14.10 : *********466484 : 125 : 008.75

14 : 15.11 : *********427801 : 001 : 000.07

16 : 22.50 : *********765859 : 225 : 015.75

17 : 07.00 : *********454555 : 021 : 001.47

18 : 23.29 : ******890044406 : 368 : 552.00�890044406 890044444 765859 466484 454555

19 17 11

3.56 0.01�Yra skambučių, kurie prasideda vieną parą, o baigiasi kitą parą.��6�5

10 : 21.50 : *********766666 : 030 : 002.10

11 : 14.10 : *********776666 : 125 : 008.75

14 : 15.11 : *********746666 : 001 : 000.07

16 : 03.10 : *********746666 : 015 : 001.05

18 : 07.00 : **********82209 : 021 : 010.05�82209 776666 766666 746666

11 10 18

54.36 0.32�Iš viso kalbėta keturiais numeriais, todėl nesusidaro reikiamas numerių skaičius��7�20

10 : 07.50 : **8372585741252 : 001 : 002.00

10 : 14.10 : *********466666 : 125 : 008.75

10 : 18.11 : *********422222 : 001 : 000.07

12 : 03.10 : *********765859 : 015 : 001.05

12 : 07.00 : *********454555 : 021 : 001.47

12 : 11.11 : ******890044406 : 091 : 025.00

19 : 22.15 : ******827201224 : 001 : 001.56

21 : 13.30 : *********748945 : 004 : 003.22

21 : 19.35 : *********454555 : 001 : 000.07

21 : 20.01 : *********223322 : 122 : 008.54

22 : 21.50 : *********366666 : 001 : 000.07

22 : 21.52 : *********366666 : 001 : 000.07

22 : 21.53 : *********366666 : 001 : 000.07

22 : 21.54 : *********366666 : 001 : 000.07

22 : 21.55 : *********366666 : 001 : 000.07

22 : 22.02 : *********366666 : 122 : 025.00

25 : 22.15 : ******827201224 : 001 : 001.56

29 : 13.30 : *********748945 : 001 : 000.07

29 : 13.55 : *********454555 : 040 : 016.80

29 : 19.01 : *********223322 : 001 : 000.07�366666 890044406 454555 466666 223322

10 12 21

68.49 6.02�Yra keturios paros kada prakalbėta daugiausiai. Į atsakymą gali įeiti ir 22-a para��8�13

10 : 09.50 : *********729601 : 010 : 000.70

10 : 10.55 : *********729602 : 008 : 000.56

10 : 11.00 : *********729605 : 004 : 000.32

10 : 11.15 : *********729606 : 010 : 000.70

10 : 11.37 : *********729607 : 008 : 000.56

10 : 11.49 : *********729209 : 002 : 000.14

10 : 12.00 : *********729209 : 002 : 000.14

10 : 12.05 : *********729209 : 002 : 000.14

10 : 12.16 : *********729209 : 001 : 000.07

10 : 12.19 : *********729209 : 001 : 000.07

10 : 12.29 : *********729209 : 001 : 000.07

11 : 08.00 : *********729209 : 001 : 000.07

12 : 07.59 : *********729219 : 008 : 000.56�729601 729606 729209 729607 729602

10 12 11

100.00 6.83�Yra šeši numeriai kuriais prakalbėta daugiausiai, į atsakymą gali įeiti ir numeris 729219; nėra tarptautinių pokalbių���157. Žaidimas su šaške. Įsigilinę į sąlygą pamatysite, kad žaidimo tikslas – pasiekti paskutinį kvadratėlį per kuo mažesnį žingsnių skaičių. Tiesa, sąlyga nėra visai aiški. Tik iš sąlygoje esančio paveikslo galima suprasti, kad žaidimas nutraukiamas atlikus pirmąjį ėjimą kurio metu buvo aplankytas n-asis kvadratėlis, o ne atlikus žingsnį, kuriuo buvo pasiektas n-ąsis kvadratėlis ar atlikus visus ėjimus iki tolesnio kauliuko metimo. T. y. žaidimą galima užbaigti, nors kvadratėlyje, kuriame stovi šaškė įrašytas ne nulis.

Teisingas, bet neefektyvus sprendimas būtų toks: perrinkti visus kauliuko metimus. Algoritmas išeitų gan paprastas ir trumpas. 

Yra efektyvesnis sprendimas. Sudarkime dvi lenteles sk[i] ir eita[i] (1 ( i ( n). Pirmoji lentelė rodys kiek mažiausiai ėjimų reikia padaryti norint pasiekti i-tąjį kvadratėlį, antroji – iš kurio kvadratėlio ateita į i-tąjį kvadratėlį, einant trumpiausiu keliu.

Imkime pavyzdį, sutampantį su pirmu testu: lentoje nupiešta 10 kvadratėlių. Ant jų užrašyta:

Kvadratėlio nr.�1�2�3�4�5�6�7�8�9�10��Skaičius ant kvadratėlio�0�5�2�-1�1�3�1�-2�0�3��

Kol neaplankėme nei vieno kvadratėlio, šaškė stovi pirmame kvadratėlyje, taigi tam kvadratėliui pasiekti reikia 0 ėjimų.

Metamas kauliukas. Gali iškristi 1, 2, 3...6 akys.

Jei iškrito viena akis, pakliuvome į antrą kvadratėlį per 1 žingsnį. Jei iškrito dvi akys, pakliuvome į trečią kvadratėlį per 2 žingsnius. Ir t. t.

Papildome lenteles sk ir eita:

Kvadratėlio nr.�1�2�3�4�5�6�7�8�9�10��Skaičius ant kvadratėlio�0�5�2�-1�1�3�1�-2�0�3��mažiausias ėjimų skaičius (sk)�0�1�2�3�4�5�6�–�–�–��iš kur ateita (eita)�1�1�1�1�1�1�1�–�–�–��

Toliau bandome eiti iš kiekvieno kvadratėlio, tikėdamiesi pasiekti kurį nors langelį per mažesnį dar mažesnį žingsnių skaičių. Jei pavyko sumažinti žingsnių skaičių iki nors vieno kvadratėlio, vėl bandome eiti iš kiekvieno kvadratėlio: gal pavyks rasti dar trumpesnį kelią iki kurio nors kvadratėlio.

Jei žingsnių skaičiaus sumažinti nepavyko – darbas baigiamas.



Parodysime, kaip toliau pildomos lentelės mūsų pavyzdžio atveju. Antrame kvadratėlyje įrašyta 5. Taigi, iš jo pakliūvame į septintą per 6 žingsnius (mat norėdami pasiekti antrą kvadratėlį padarėme 1 žingsnį, o iš jo paėjome dar per 5 kvadratėlius). Analogiškai iš trečio kvadratėlio pakliūvame į penktą, per 4 žingsnius, iš ketvirto į trečią per 4 žingsnius, iš penkto į šeštą per 5 žingsnius. Ligi šiol neradome trumpesnio kelio nei į vieną kvadratėlį.

Iš šešto kvadratėlio pakliūvame į devintą per 8 žingsnius, iš septinto į aštuntą – per 7 žingsnius. Papildome lenteles:

Kvadratėlio nr.�1�2�3�4�5�6�7�8�9�10��Skaičius ant kvadratėlio�0�5�2�-1�1�3�1�-2�0�3��mažiausias ėjimų skaičius (sk)�0�1�2�3�4�5�6�7�8�–��iš kur ateita (eita)�1�1�1�1�1�1�1�7�6�–��

Kadangi pasiekėme du naujus kvadratėlius, lentelę peržiūrime iš naujo. Kaip ir ankstesniu atveju, eidami iš pirmo-septinto kvadratėlių, negausime nieko naujo. Iš aštunto kvadratėlio pakliūsime į šeštą per 7 žingsnius. Devintame kvadratėlyje teks mesti kauliuką. Išmetę 1 akį, pakliūsime į dešimtą kvadratėlį per 9 žingsnius, išmetę 2-6 akis, pakliūsime į pirmą-penktą kvadratėlius. Papildome lenteles:

Kvadratėlio nr.�1�2�3�4�5�6�7�8�9�10��Skaičius ant kvadratėlio�0�5�2�-1�1�3�1�-2�0�3��mažiausias ėjimų skaičius (sk)�0�1�2�3�4�5�6�7�8�9��iš kur ateita (eita)�1�1�1�1�1�1�1�7�6�9��

Kadangi lentelės buvo keistos, dar kartą bandome eiti iš visų kvadratėlių.

Neberandame trumpesnio kelio ir darbą baigiame. Beje, papildžius lenteles užtektų bandyti eiti ne iš visų kvadratėlių, o tik iš tų kurių reikšmės buvo pakeistos.



const MAX = 10000;   { maksimalus kvadratėlių skaičius }

type lentelė = array [1..MAX] of integer;

       { lentelė, rodanti per kiek mažiausiai ėjimų galima pasiekti }

       { kiekvieną langelį }



procedure žaisti (n: integer; kv: lentelė;

  { randama keliais mažiausiais ėjimais pasiekiamas kiekvienas langelis }

                  var eita: lentelė);

  var sk: lentele; { mažiausias ėjimų skaičius per kurį pasiekiamas }

                   { kiekvienas langelis }

      kur, i, j: integer;

      trumpesnis: boolean;

begin

  sk[1] := 0; eita[1] := 0;

  for i := 2 to n do

    begin

      eita[i] := 0;   { inicializuojame lenteles }

      sk[i] := maxint div 2; { didelis skaičius – langelis nepasiektas }

    end;

  trumpesnis := true;

  while trumpesnis do  { kol radome trumpesnį kelią }

    begin

      trumpesnis := false;

      for i := 1 to n do  { bandysime eiti iš i-tojo langelio }

        if kv[i] <> 0 { jei kvadratėlyje yra ne nulis }

        then begin

               { (i + kv[i]) – į tokį langelį pakliūsime }

               { (sk[i+kv[i]]) – tiek žingsnių reikėjo anksčiau }

               { (sk[i] + kv[i]) – tiek žingsnių reikia dabar }

               if i + kv[i] > n { pakliūsime į šį langelį }

                  then kur := n { neverta vėl eiti į pirmus langelius }

                  else kur := i + kv[i];

               if kur > 0 { iš pirmo langelio neiname atgal }

                  then 

                    if sk[i] + abs(kv[i]) < sk[kur]

                      then begin

                            { iš paskutinio langelio einame į pirmą }

                            sk[kur] := sk[i] + abs(kv[i]);

                            eita[kur] := i;

                            trumpesnis := true;

                           end

             end

        else { jei i-tame kvadratėlyje įrašytas nulis }

          for j := 1 to 6 do { metame kauliuką }

            if i + j <= n { iš paskutinio langelio neverta eiti į pirmą }

               then if sk[i] + j < sk[i+j]

                       then begin

                              sk[i+j] := sk[i] + j;

                              eita[i+j] := i;

                              trumpesnis := true;

                             end

    end; { while trumpesnis do }

end; { žaisti }



Užpildžius lenteles reikia gauti kauliuko akių seką. Vėl pasiremsime tuo pačiu pavyzdžiu. Į dešimtą kvadratėlį atėjome iš devinto. Kadangi devintame metamas kauliukas, tai reiškia kad buvo iškritusi 1 akis. Į devintą pakliuvome iš šešto langelio (jame kauliukas nemetamas), o į šeštą – iš pirmo. Jame kauliukas metamas, taigi buvo iškritusios 5 akys.

Ieškoma kauliuko metimų seka: 5 1.



procedure rašyk (var kv, eita: lentelė; 

                 n: integer; var f: text);

{ rekursinė procedūra, spausdinanti rezultatus }

  var i: integer;

begin

  if n > 1

     then if kv[eita[n]] = 0

             then begin

                    rašyk (kv, eita, eita[n], f);

                    writeln (f, n-eita[n]);

                  end

              else rašyk (kv, eita, eita[n], f);

end; { rašyk }





Vyresniųjų grupė

158. Skaičiai. Pateiksime algoritmą vieną skaičiaus užrašą pakeičiantį jo skaitine reikšme. Norintys nesunkiai papildys programą, kad būtų keičiami visi byloje esantys skaičiai. Uždavinio sprendimui bus būtinos dvi procedūros. Pirmoji jų skaityti perskaito iš bylos vieną žodį. 

procedure skaityti (var b: text; var z: string);

  { skaitomas vienas žodis z iš bylos b }

  { perskaičius paskutinį eilutės žodį eoln(b) = true }

  var s: char;

begin

  z := ''; s := ' ';

  { praleidžiami tarpai }

  while not eoln(b) and not eof(b) and (s = ' ') do

    read(b, s);

  { perskaitomas pats žodis }

  if s <> ' ' then z := z + s;

  while not eoln(b) and not eof(b) and (s <> ' ') do

    begin

        read(b, s);

        if s <> ' ' then z := z + s

      end

end; { skaityti }



Antroji procedūra skaitvk patikrina ar skaitvardis yra gramatiškai teisingas ir grąžina jo kodą (skaitinę)reikšmę. Kiekinių skaitvardžių užrašuose vartojami vienaskaitos vardininko linksnyje esantys skaitvardžiai (vienas, du, …, dešimt, vienuolika, dvylika, …devyniolika, dvidešimt, tridešimt, …devyniasdešimt, šimtas, tūkstantis). Ne vienaskaitos vardininko linksnyje gali būti vartojami tik šie skaitvardžiai: šimtai (du šimtai), tūkstančiai (du tūkstančiai), tūkstančių (dešimt tūkstančių). Šių trijų žodžių kodai skirsis nuo jų reikšmių.

function skaitvk (skaitv: string): integer;

  { skaitvardis –> skaitvardžio kodas (beveik visada – reikšmė) }

  const skaitvsk = 32; { skaitvardžių skaičius }

        lent: array [1..skaitvsk] of 

                 record { skaitvardžių kodai }

                    s: string;  { skaitvardis }

                    k: integer  { kodas }

                 end =          { (dažnai – ir reikšmė) }

      ((s: 'vienas' ;         k: 1),

       (s: 'du';              k: 2),

       (s: 'trys';            k: 3),

       (s: 'keturi';          k: 4),

       (s: 'penki';           k: 5),

       (s: 'šeši';            k: 6),

       (s: 'septyni';         k: 7),

       (s: 'aštuoni';         k: 8),

       (s: 'devyni';          k: 9),

       (s: 'dešimt';          k: 10),

       (s: 'vienuolika';      k: 11),

       (s: 'dvylika';         k: 12),

       (s: 'trylika';         k: 13),

       (s: 'keturiolika';     k: 14),

       (s: 'penkiolika';      k: 15),

       (s: 'šešiolika';       k: 16),

       (s: 'septyniolika';    k: 17),

       (s: 'aštuoniolika';    k: 18),

       (s: 'devyniolika';     k: 19),

       (s: 'dvidešimt';       k: 20),

       (s: 'trisdešimt';      k: 30),

       (s: 'keturiasdešimt';  k: 40),

       (s: 'penkiasdešimt';   k: 50),

       (s: 'šešiasdešimt';    k: 60),

       (s: 'septyniasdešimt'; k: 70),

       (s: 'aštuoniasdešimt'; k: 80),

       (s: 'devyniasdešimt';  k: 90),

       (s: 'šimtas';          k: 100),

       (s: 'šimtai';          k: 102), { kodas <> reikšmė }

       (s: 'tūkstantis';      k: 1000),

       (s: 'tūkstančiai';     k: 1002), { kodas <> reikšmė }

       (s: 'tūkstančių';      k: 1010)); { kodas <> reikšmė }



  var i: integer;

begin

  skaitvk := 0;

  for i := 1 to skaitvsk do

      if lent[i].s = skaitv then skaitvk := lent[i].k

end; { skaitvk }



Užrašo pavertimas skaitine reikšme remiasi tuo, kad prieš ir po žodžio tūkstantis (čiai, čių) gali eiti įprasto triženklio skaičiaus užrašas. Formuojamas skaičius suskaidomas į tūkstančių ir vienetų komponentus. Pavyzdžiui: trys šimtai du tūkstančiai du šimtai dvidešimt –> 302 ir 220. Sujungę gausime 302220. Suformavus pirmą komponentą ieškoma arba užrašo pabaigos arba žodžio tūkstantis (čiai, čių). Komponento skaitinė reikšmė formuojama iš kairės į dešinę. 



procedure formuotisk(var prad: text; var visassk: longint);

  { skaitvardžiai –> skaičius }

  var skaitv: string;      { skaitvardis }

      kodas: integer;      { skaitvardžio kodas }

      tkst, vnt: longint;  { skaičiaus komponentai po 3 skaitmenis: }

                           { tūkstančiai (xxx000) ir vienetai (000xxx) }

      klaida: boolean;



begin

 tkst := 0; vnt := 0; klaida := false;

 while not eoln(prad) do

  begin

   skaityti(prad, skaitv);

   kodas := skaitvk(skaitv);

   case kodas of

    0:      klaida := true;

    1..9:   if vnt mod 10 = 0

               then vnt := vnt + kodas

               else klaida := true;

    10..90: if (vnt mod 100) = 0

               then vnt := vnt + kodas

               else klaida := true;

    100:    if vnt <= 1  { (1) šimtas }

               then vnt := 100

               else klaida := true;

    102:    if vnt in [2..9] { 2 šimtai }

               then vnt := vnt * 100

               else klaida := true;

    1000:   begin

             if vnt = 0 then vnt := 1; { nebuvo „vienas“ }

             if (tkst = 0)

                 and (vnt mod 10 = 1)   { 1 tūkstantis }

                 and (vnt div 10 mod 10 <> 1) { bet ne 11..19 tūkst. }

                 then begin 

                        tkst := vnt; 

                        vnt := 0 

                      end

                 else klaida := true

            end;

    1002:   if (tkst = 0)

               and ((vnt mod 10) in [2..9]) { ... 2..9 tūkst., }

               and (vnt div 10 mod 10 <> 1) { bet ne 11..19 t. }

               then begin 

                      tkst := vnt; 

                      vnt := 0 

                    end

               else klaida := true;

    1010:   if (tkst = 0)

               and (vnt > 0)

               and ((vnt mod 100) in [11..19]) { ...11..19 tūkst. }

               or ((vnt mod 100 div 10 in [0..9])

               and (vnt mod 10 = 0)) { 10, 20, ..., 100 t. }

               then begin 

                      tkst := vnt; 

                      vnt := 0 

                    end

               else klaida := true

   end; { case }

  end; { while }

  if klaida 

     then visassk := 0

     else visassk := tkst*1000 + vnt;

  readln(prad)

end; { formuoti_sk }





159. Žaidimas „Minų laukas“. Aiškindami sprendimą piešiniuose baltai žymėsime atvertus langelius, pilkai – užvertus.

Jei užvertą langelį gaubia tik užverti langeliai, niekaip negalėsime nustatyti kas jame yra (159.2 pav.). To nepavyks padaryti net jei atversime visus kaimyninius langelius (159.3 pav.).

1�(�2��0��1����2��1��?��(��2����2��2��2�0�0��159.2 pav.



1�(�2��0��1��(��2��1��?�(�(��2�(���2��2�(�2�0�0��159.3 pav.



Taip yra todėl, kad atvėrę langelį galime pasakyti ar jame yra mina, ar ne. Tačiau negalime pasakyti koks skaičius jame turėtų būti įrašytas. 

Taigi, galima atverti tik langelius, gretimus langeliams, kuriuose įrašyti skaičiai. 

Yra tik du atvejai, kada galima iš karto nustatyti, kas yra aplink laukelį su skaičiumi: 

Jei langelyje įrašytas skaičius lygus aplink jį esančių užminuotų ir uždengtų laukelių sumai, tai uždengtuosiuose laukeliuose yra minos; (159.4 pav.)

Jei langelyje įrašytas skaičius lygus aplink laukelį esančių minų skaičiui, tai likę uždengti langeliai yra tušti. (159.5 pav.);

(�(�(��4�5�2��(�(�2��159.4 pav.



�(�(��2�3�4���(�2��159.5 pav.

Sudarome langelių, kuriuose įrašyti skaičiai, sąrašą ir kiekvienam jų patikriname šiuos du variantus. Jei kuris nors tinka – atveriami kaimyniniai langeliai ir įrašomos reikšmės (užminuotas ar laisvas), o pats langelis pašalinamas iš sąrašo. Atidengus laukelius, gali pasikeisti situacija aplink kitus langelius su skaičiais (159.6 pav.). Todėl sąrašą reikia tikrinti tol, kol nebebus padaryta pakeitimų, arba sąrašas liks tuščias. Pastaruoju atveju jau būsime suradę sprendinį. 



�1�1�1����(�1���4��1���(����159.6.a pav. Aplink langelį su skaičiumi 4 yra dvi minos ir keturi užverti langeliai. Iš karto negalime pasakyti kuriuose iš užvertų langelių yra minos.



�1�1�1����(�1���4��1���(����159.6.b pav. Atvėrus keletą kitų langelių, aplink langelį su skaičiumi 4 yra dvi minos ir du užverti langeliai. Jau galime pasakyti, kad užvertuose langeliuose yra minos.

Žemiau pateikta procedūra, kurios atverčia tokius langelius. Prieš ją rasite keletą pagalbinių funkcijų bei procedūrų, kurių veikimas lengvai suprantamas. Jose naudojami globalūs kintamieji n bei lau. 

const DYDIS = 75; { maksimalus žaidimo laukas}

type laukas = array [1..DYDIS, 1..DYDIS] of char;



{ žemiau aprašyti globalieji kintamieji }

var n: integer;  { lauko dydis }

    lau,         { žaidimo laukas }



function lauke (x, y: integer): boolean;

begin

  lauke := (x >= 1) and (x <= n) and (y >= 1) and (y <= n);

end; { lauke }



function minos (x, y: integer ): integer;

  { kiek minų supa langelį (x, y) }

  var i, sk: integer;

begin

  sk := 0;

  for i := 1 to 8 do

    if lauke (x + d[i].x, y + d[i].y)

       then if lau[x + d[i].x, y + d[i].y] = 'M'

               then sk := sk + 1;

  minos := sk

end; { minos }



function užv (x, y: integer): integer;

  { kiek užvertų langelių supa duotą langelį }

  var i, sk: integer;

begin

  sk := 0;

  for i := 1 to 8 do

    if lauke (x + d[i].x, y + d[i].y)

       then if lau[x + d[i].x, y + d[i].y] = 'U'

               then sk := sk + 1;

  užv := sk

end; { užv }



function ska (c: char): integer;

  { simbolinį korektišką skaitmens užrašą paverčia skaičiumi }

begin

  ska := ord(c) - ord('0');

end; { ska }



procedure keisti (x, y: integer; simb: char; var lau: laukas);

  { visus užvertus langelius esančius apie langelį (x, y), pakeičiame nurodytu simboliu simb }

  var i, xx, yy : integer;

begin

  for i := 1 to 8 do

    begin

      xx := x + d[i].x; yy := y + d[i].y;

      if lauke (xx, yy) { jei neišeinama už lauko ribų }

         then if lau[xx, yy] = 'U'

                 then lau[xx, yy] := simb;

    end;

end; { keisti }



Dabar jau galime pateikti procedūrą atversti. Visi lauko langeliai, kuriuose yra skaičiai, įrašyti sąraše tikrinti. Tai padaryta prieš iškviečiant šią procedūrą.

const DYDIS = 75; { maksimalus žaidimo laukas}

type laukas = array [1..DYDIS, 1..DYDIS] of char;

     sąrašas = array [1..DYDIS*DYDIS] of record

                                           x, y : integer

                                         end;



procedure atversti (var ilg: integer; var tikrinti: sąrašas;

                    var lau: laukas);

  { atverčiami langeliai, kuriuos galima atversti be perrinkimo }

  var pirma, antra,

      keista: boolean;  { ar padaryti pakeitimai žaidimo lauke }

      i: integer;

begin

  keista := true;

  while keista do   { ieškoti tol, kol buvo padaryta pakeitimų }

    begin

      keista := false;

      i := 1;

      while i <= ilg do

        with tikrinti[i] do

          begin

            { ar tenkinama pirma ar antra sąlygos }

            pirma := ska(lau[x, y]) = minos(x, y) + užv(x, y);

            antra := ska(lau[x, y]) = minos (x, y);

            if pirma

               then keisti (x, y, 'M', lau)  { pirma sąlyga }

               else if antra

                       then keisti (x, y, 'L', lau); { antra sąlyga }

            if pirma or antra

               then begin

                      keista := true;

                      { iš sąrašo pašalinamas elementas i }

                      tikrinti[i] := tikrinti[ilg];

                      ilg := ilg - 1;

                    end

                else i := i + 1;

          end

    end;

end; { atversti }



Jei sąrašas dar netuščias ir daugiau nebegalime atverti nė vieno langelio, teks perrinkti visus galimus variantus.

Visus langelius (arba galime sakyti, visą lentą) suskirstome į grupes. Jei langeliai priklauso skirtingoms grupėms, tuomet jų reikšmes galime perrinkti nepriklausomai (159.7 pav.) 



�1�2�3�4��1�2���1��2����2��3������4�1��2���159.7 pav. Padalijimas į grupes parodytas storesnėmis linijomis. Aiškumo dėlei eilutės bei stulpeliai sunumeruoti.

Kaip padalinti į tokias grupes? Patogiausia dalinti pradedant nuo langelių, kuriuose įrašyti skaičiai. Du langeliai su skaičiais laikomi susiję, jei jie turi bendrą kaimyninį langelį. Visi langeliai, gretimi langeliui su skaičiumi priklauso tai pačiai grupei kaip ir tas langelis su skaičiumi.

Grįžkime prie 159.7 pav. Imkime langelį (1, 1). Jį paskirkime į pirmąją grupę. daugiau nėra nė vieno langelio su skaičiumi, kuris turėtų tą patį kaimyninį langelį, kaip ir (1, 1). Visi jam kaimyniniai užverti langeliai taip paskiriami į pirmąją grupę. Taigi, pirmajai grupei priklausys langeliai (1, 1), (1, 2), (2, 1) bei (2, 2). Trys iš jų užverti. Jų reikšmės (yra mina, nėra minos, nežinomi visiškai nepriklauso nuo to, kas yra kituose šiai grupei nepriklausančiuose langeliuose.

Imkime kitą langelį su skaičiumi: (1, 4). Šis langelis ir langelis (2, 4) turi net du bendrus kaimynus: (1, 3) ir (2, 3). Taigi, jie pakliūva į tą pačią grupę. Langelis (2, 4) ir langelis (4, 3) turi du bendrus kaimynus: (3, 3) ir (3, 4). Tad ir langelis (4, 3) paskiriamas į tą pačią grupę. Langeliai (4, 3) ir (4, 1) turi du bendrus kaimynus: (4, 2) ir (3, 2). Taigi ir langelis (4, 1) paskiriamas į tą pačią grupę. Matome, kad visi langeliai su skaičiais, išskyrus (1, 1) priklauso antrajai grupei kartu su juos gaubiančiais langeliais užvertais langeliais.

Žemiau pateikta procedūra, kuri atskiria vieną grupę iš langelių su skaičiais, esančių sąraše tikrinti. Juos gaubiantys užverti langeliai taip pat priklausys tai pačiai grupei, tačiau patogiau jų į lentelę, kurioje saugomi grupei priklausantys langeliai, neįtraukti. T. y. į lentelę įrašomi tik tos grupės langeliai su skaičiais.

Langeliai, atskirti į tą grupę, iš sąrašo tikrinti išbraukiami.

procedure grupuoti (var sk, ilg: integer; { rastųjų langelių skaičius }

                    var grupė, tikrinti: sąrašas);

  { ieškoma langelių, susijusių su paskutiniu sąrašo tikrinti langeliu; 

   rastieji langeliai pašalinami iš sąrašo tikrinti }

  var i, j, k: integer;

begin

  { ieškosime langelių, susijusių su paskutiniuoju sąrašo langeliu }

  sk := 1;

  grupė[sk] := tikrinti[ilg];

  ilg := ilg - 1; { laneglis pašalinamas iš sąrašo }

  i := 1;

  while i <= sk do { ar sąrašo grupė langelis }

    begin

      j := 1;

      while j <= ilg do { susijęs su sąrašo tikrinti langeliu }

        begin

          if (abs(tikrinti[j].x-grupė[i].x) < 3) and { jei atstumas }

             (abs(tikrinti[j].y-grupe[i].y) < 3)     { tinkamas }

          then begin

                 sk := sk + 1;             { įrašome naują susijusį }

                 grupė[sk] := tikrinti[j]; { langelį į sąrašą grupė }

                 tikrinti[j] := tikrinti[ilg]; { ir pašaliname iš sąrašo }

                 ilg := ilg - 1;               { tikrinti }

                 j := j - 1;

               end;

          j := j + 1

        end;

      i := i + 1;

    end

end; { grupuoti }



Turėdami susijusių langelių grupę, jau galima perrinkti visus galimus atvejus. T. y. į kiekvieną grupės užvertą langelį pirma bandoma padėti miną, o po to palikti jį laisvą. Padėjus miną (ar palikus jį laisvą) patikrinama, ar nesusidarė nekorektiška situacija kitam tos pačios grupės langeliui su skaičiumi. T.y. ar visi kiti grupės langeliai su skaičiais turi pakankamai gretimų langelių į kuriuos galės įrašyti minas ir ar nė vieno langelio su skaičiumi nesupa didesnis nei reikia minų skaičius. Jei gavome nekorektišką situaciją, grįžtame vieną žingsnį atgal. Jei situacija korektiška – tęsiame perrinkimą kol gauname sprendinį. Jį įsimename.

Tuomet grįžtame per žingsnį atgal ir toliau atliekame perrinkimą ieškodami kitų sprendinių. Taip randame visus galimus sprendinius. 

– Jei užvertas langelis visuose sprendiniuose yra tuščias, tuomet jį atveriame. 

– Jei tame langelyje visuose sprendiniuose yra mina – langelyje paliekame miną. 

– Jei viename (ar keliuose) sprendiniuose konkrečiame langelyje yra mina, o kitame (ar keliuose kituose) sprendinyje tame pačiame langelyje minos nėra, tada negalime vienareikšmiškai pasakyti kas turi būti tame langelyje. (159.8 pav.)

�1�2�3�4��1������2�2���1��3������4�1�1�1�1��159.8.a pav.





�1�2�3�4��1������2�2�(��1��3�(���(��4�1�1�1�1��159.8.b pav. Pirmasis sprendinys





�1�2�3�4��1�(�����2�2���1��3�(���(��4�1�1�1�1��159.8.c pav. Antrasis sprendinys





�1�2�3�4��1��(����2�2���1��3�(���(��4�1�1�1�1��159.8.d pav. Trečiasis sprendinys



�1�2�3�4��1�N�N����2�2�N��1��3�(���(��4�1�1�1�1��159.8.e pav. Galutinis sprendinys



Imkime pavyzdį, pateiktą 159.8 pav. Pirmame sprendinyje langelyje (1, 1) minos nėra, o antrame nėra. Langelis lieka nežinomas. Langeliuose (3, 1) ir (3, 4) visuose sprendiniuose minos yra, taigi ir galutiniame sprendinyje mina lieka. Nė viename sprendinyje langelyje (1, 4) minos nėra. Galutiniame sprendinyje minos ten taip pat nebus. 

const DYDIS = 75; { maksimalus žaidimo laukas}

      d: array [1..8] of record { aštuonios kryptys } 

                            x, y: integer;

                         end = ((x: -1; y: -1), (x: -1; y:  0),

                                (x: -1; y:  1), (x:  0; y: -1),

                                (x:  0; y:  1), (x:  1; y: -1),

                                (x:  1; y:  0), (x:  1; y:  1));

type laukas = array [1..DYDIS, 1..DYDIS] of char;

     sąrašas = array [1..DYDIS*DYDIS] of record

                                           x, y : integer

                                         end;



{ žemiau aprašyti globalieji kintamieji }

var n: integer;  { lauko dydis }

    lau,         { žaidimo laukas }

    spr: laukas; { rastasis sprendinys }



    ilg: integer;      { tikrintinų langelių sąrašo ilgis }

    tikrinti: sąrašas; { tikrintinų langelių sąrašas }



    sk: integer;       { nagrinėjamos grupės langelių sąrašo ilgis }

    grupė: sąrašas;    { nagrinėjama grupė }





procedure minų_laukas (var spr: laukas);

  

  procedure atnaujinti_spr (const lau: laukas; var spr: laukas);

    { atnaujinamas galutinis sprendinys, esantis lentelėje spr ; lau – naujai rastas sprendinys }

    var i, j : integer;

  begin

    for i := 1 to n do

      for j := 1 to n do

        case spr[i, j] of

          'U': if lau[i, j] in ['M', 'L']

                  then spr[i, j] := lau[i, j];

          'M': if lau[i, j] <> 'M'        { langelyje gali būti mina, }

                  then spr[i, j] := 'N';  { bet gali jos ir }

          'L': if lau[i, j] <> 'L'        { nebūti }

                  then spr[i, j] := 'N';

          'N', '0'..'9': ; { nekeisti nieko }

        end;

  end; { atnaujinti_spr }



  function gerai (x, y: integer): boolean;

    { ką tik atvertas langelis (x, y); patikrinama, ar viskas gerai }

    var taip: boolean;

        i, xx, yy: integer;

  begin

    taip := true;

    for i := 1 to 8 do

      begin

        xx := x+d[i].x; yy := y+d[i].y;

        if lauke (xx, yy)

         then if lau[xx, yy] in ['0'..'9']

               then taip := taip and 

                    (minos(xx, yy) <= ska(lau[xx, yy])) and

                    (ska(lau[xx,yy]) <= minos(xx,yy)+uzv(xx,yy))

      end;

    gerai := taip;

  end; { gerai }



  procedure ieškoti (nr, kaim: integer);

    { dėliojamos minos apie langelį, kurio numeris masyve grupė yra nr }

    var x, y, xx, yy: integer;

  begin

   if nr > sk

     then atnaujinti_spr (lau, spr) { rastas sprendinys }

   else begin

          x := grupė[nr].x; y := grupė[nr].y;

          if užv (x, y) = 0 { jei visi langeliai atverti }

             then ieškoti (nr+1, 1)

          else begin

                 xx := x + d[kaim].x; yy := y + d[kaim].y;

                 if lauke(xx, yy) then

                   if lau[xx, yy] = 'U' then

                    begin

                      { dedame miną }

                      lau[xx, yy] := 'M';

                      { patikriname, ar viskas gerai }

                      if gerai (xx, yy) 

                         then ieškoti (nr, kaim+1);

                      { nededame minos }

                      lau[xx, yy] := 'L';

                      if gerai (xx, yy) 

                         then ieškoti (nr, kaim+1);

                      { atstatome reikšmę }

                      lau[xx, yy] := 'U';

                    end

                   else ieškoti (nr, kaim+1) { jei ne užvertas langelis }

                 else ieškoti (nr, kaim+1); { jei už lauko ribų }

               end;

        end;

  end; { ieškoti }





  var i, j: integer;

begin { minų_laukas }

  { langelius su skaičiais, įrašome į masyvą tikrinti }

  ilg := 0;

  for i := 1 to n do

    for j := 1 to n do

      if lau[i, j] in ['0'..'9']

         then begin

                ilg := ilg + 1;

                with tikrinti[ilg] do

                  begin

                    x := i; y := j;

                  end;

              end;

  { atveriame langelius, kuriems nereikia perrinkimo }

  atversti (ilg, tikrinti, lau);

  { ieškosime sprendinio }

  spr := lau;

  while ilg <> 0 do { kol sąrašas tikrinti nėra tuščias }

    begin

      grupuoti (sk, ilg, grupė, tikrinti); { atskiriame grupę }

      ieškoti (1, 1);                    { ir ieškome sprendinio }

      lau := spr;

    end;

  { visi likę užverti langeliai – nežinomi }

  for i := 1 to n do

    for j := 1 to n do

      if spr[i, j] = 'U'

         then spr[i, j] := 'N';

end; { minų_laukas }



Įvertinsime šio algoritmo vykdymo laiką blogiausiu atveju. Sakykime, visi lentos langeliai priklauso tai pačiai grupei ir be perrinkimo negalima atversti langelių. Apie kiekvieną atverstą langelį gali būti iki aštuonių užverstų langelių. Pakankamai blogu atveju susijusių užverstų langelių skaičius gali siekti 4000, į kiekvieną langelį galima įrašyti vieną iš dviejų reikšmių (yra mina arba jos nėra). Taikant grįžimo metodą vykdymo laikas auga eksponentiškai. Tad apytikriai vykdymo laikas bus 24000 eilės. 

Ne visi skaitytojai gali žinoti ką reiškia pasakymas algoritmo vykdymo laikas yra k eilės. Imkime paprasčiausią veiksmą, pavyzdžiui sakinį: a :( 1. Jį kompiuteris įvykdo per laiką t. Be abejo, t priklauso nuo kompiuterio charakteristikų, jo greičio. k eilės algoritmo ilgiausias vykdymo laikas bus apytikriai lygus kt. 

Gali kilti klausimas, kaip gali vykdymo laiko eilė nusakyti algoritmo atlikimo greitį. Juk kompiuterių greičiai gali skirtis dešimtis ar net šimtus kartų. Pamėginkime užpildyti žemiau esančią lentelę. 

Algoritmo vykdymo laiko eilė�Vykdymo greitis��215�0,06 sek.��217�0,16 sek.��218�0,28 sek.��219�0,55 sek.��220�1,10 sek.��221�2,12 sek.��222�4,35 sek.��226�1 min 12 sek.��Šioje lentelėje algoritmo vykdymo eilė auga eksponentiškai. Kaip matome laipsnio rodikliui padidėjus vienetu, vykdymo laikas pailgėja maždaug dvigubai. Pratęskime lentelę.

Algoritmo vykdymo laiko eilė�Vykdymo greitis��230�20 min.��234�5 val.��238�3 paros��242�1,5 mėn.��245�1 metai��255�1000 metų��Be abejo, paskutiniosios šios lentelės eilutės buvo užpildytos remiantis teoriniais apskaičiavimais, o ne praktiniais bandymais... Ir pasiekėme tik 255. Iki 24000 dar labai toli...

Matome, kad algoritmo vykdymo laiko eilė labai daug pasako apie jo galimą vykdymo laiką. Pavyzdžiui, jei algoritmo vykdymo laiko eilė lygi 226, tai viename kompiuteryje jis bus vykdomas 5 sekundes, kitame – 1 minutę, dar kitame – 1 valandą. Visais trimis atvejais įmanoma sulaukti rezultato. O kompiuterio greičio charakteristika, kurią pavadinome t, nulems kiek laiko teks laukti rezultato.

Jei algoritmo vykdymo laiko eilė yra 255, tai viename kompiuteryje jis bus vykdomas 1000 metų, kitame – 1500 metų, o naujausiuose kompiuteriuose – 100 metų... Šiuo atveju net ir greičiausias šiuolaikinis kompiuteris nepadės... Jei kompiuteriai pagreitėtų dar kokius 1000 kartų, uždavinys taptų išsprendžiamas.

Samprotavimus apie uždavinio su minų lauku sprendimą, kurio vykdymo laiko eilė lygi maždaug 24000, reiktų užbaigti mintimi, kad parinkus pakankamai sudėtingus testus, atsakymo sulaukti nepavyks. 

